



«Solitaire» for the Commodore 64 (Complete type-in listing in BASIC V2)



**An exclusive 2021 Give-Away by
Roman Werner (🐦 @RomWer)**

Thanks! for following me on Twitter

!- Note: Curly brackets {} represent special screen code characters (BASTEXT PETSCII Tokens)
!- For BASTEXT reference see: https://www.c64-wiki.com/wiki/PETSCII_Codes_in_Listings
!- For checksum tool see : <https://www.c64-wiki.com/wiki/F64Summer>

```
CSUM
11E2 10 rem *****
868F 20 rem *** klondike - solitaire ***
D03C 30 rem *** for commodore mouse 1351 ***
85BB 40 rem *** (c) 2021 by @romwer ***
70C5 50 rem *** roman.werner@gmail.com ***
1EC1 60 rem *****
E08D 70 poke53280,5:poke53281,5:print"{clear}{black}"spc(14)"please wait":dw=1:sh=1:so=7
7619 80 fori=0to23:poke54272+i,0:next:poke54296,so:gosub1910:gosub2090:gosub1840
E175 90 rem *** title screen ***
E39C 100 print"{clear}{black}{reverse off}{down}"spc(11)"klondike solitaire{down*6}"
0E97 110 ifhithenfori=1to4:print"{home}{down*3}"spc(5+i*5)pt$(0);:next
600D 120 print:printspc(10)"{down}{white}welcome to solitaire{down}"
6B59 130 printspc(3)"{light green}for commodore mouse 1351 in port 2{down*2}"
4454 140 printspc(13)"{yellow}f1 {light green}play random{down}":dw$=left$("s",dw-1)
DAEA 150 printspc(13)"{yellow}f3 {light green}play number{white}";:ifsd>0thenprintsd;
8D5C 160 print:printspc(14)"{down}{yellow}d {light green}draw{white}"dw"{light green}card"dw$
5334 170 printspc(14)"{down}{yellow}c {white}"mid$("hideshow",sh*4+1,4)" {light green}counter"
0E3F 180 printspc(14)"{down}{yellow}s {light green}sound {white}"mid$("offon",so/2+1,3)
92FB 190 poke251,pc:poke252,dw:poke254,so:poke2,sh:clr
1FCD 200 pc=peek(251):dw=peek(252):so=peek(254):sh=peek(2)
8DA1 210 gosub1650:fori=1to4:print"{home}{down*3}"spc(5+i*5)pt$(0);:next:poke198,0
E6CF 220 poke214,16:print:printspc(28):ifk$="{f3}"ory=8thenprintss$sr$r$"{up*2}";
BDB6 230 getk$:if(peek(j2)and1)=0thengosub1500:waitj2,1:goto230
D6F5 240 y=0:x=0:if(peek(j2)andg)=0thenx=peek(v):y=peek(v+1)-53:y=int(y/16):waitj2,g
5BFC 250 ifk$=""andy=0goto230
1C5D 260 ifx>0thenifx<122goto230
FC02 270 ifk$="{f1}"ory=7thensd=int(rnd(1)*ti)+1:goto340
DA43 280 ifk$="{f3}"ory=8thenpoke19,1:input"{white}";a$:poke19,0:sd=val(a$):on-(sd>0)goto340
4A9D 290 ifk$="d"ory=9thendw=4-dw:poke214,18:print:printspc(21)cd$(dw)
4654 300 ifk$="c"ory=10thensh=1-sh:poke214,20:print:printspc(16)sh$(sh)
D6F8 310 ifk$="s"ory=11thensso=7-so:pokes+24,so:poke214,22:print:printspc(22)v$(so)
5535 320 poke646,1:goto220
BBB8 330 rem *** new game / shuffling
4429 340 a=rnd(-sd):mv=0:poke19,0:print"{clear}{black}"spc(15)"shuffling":pokes+4,129
B5C1 350 fori=1to15:next:pokes+4,0:fori=0tohi:cd(i)=i:next:fori=hito1step-1
0430 360 a=rnd(1)*(i+1):b=cd(i):cd(i)=cd(a):cd(a)=b:ifi<30thenpokes+4,129:pokes+4,0
E01E 370 next
603B 380 rem *** show side panel
E6F1 390 print"{white}{clear}"spc(35)o$(0)d1$"{red}!"d$"{red}#"d$"{black}%"d$"{black}&"d2$u$(0)r$;
B4A5 400 print"{white}{down}"nc$r$"{down*8}{light green} @{reverse on} {reverse off}"/";
8238 410 rem *** lay out cards ***
E9F6 420 z=n:fori=0to6:lc(i)=n:next:ox=n:os=n:lc(7)=12:ifshthenpoke1422,48
5A23 430 fory=0to6:forx=yto6:z=z+1:cd=cd(z):ifx>ythencd=cd+128
5445 440 c(x,y)=cd:gosub710:next:lc(y)=y:next:e=z:pokes+4,129:pokes+4,0
019E 450 poke214,9:print:printspc(35)o$(0)pt$u$(2):ti$="000000"
DC2F 460 rem *****
1D26 470 rem *** main loop ***
FDD3 480 rem *****
2004 490 pokefe,0:cs=n:y2=n:waitfe,17:x=peek(fb)+msb(peek(fd))-23
6A96 500 x2=int(x/f):ifx2<0orx2>7orx-x2*f>30thenpokes+1:x2=n:goto630
C44F 510 rb=peek(fe)and1:lc=lc(x2):of=of(x2):y=peek(fc)-53:y2=int(y/g)+of
3A6F 520 iflc=nthenifos=12oros=25oros=38oros=51thenpokes+1:y2=n:gosub1180
A54A 530 ify2>lc+2orlc=nthenpokes+1:x2=n:y2=n:goto630
95DD 540 ify2>0thenify2>lctheny2=lc:ify>lc*g+fthenpokes+1:x2=n:y2=n:goto630
15A3 550 ifx2=7thengosub830:goto630
46B0 560 ifx2=oxandy2=oythengosub910:goto640
D4FC 570 pokes+1:ify2=12thenifof(x2)=0thengosub1380:os=n:goto490
F99F 580 ify2=8thenifof(x2)thengosub1380:os=n:goto490
5411 590 ify2>nthencs=c(x2,y2):ifcs>hiandy2<lcthencs=n
283E 600 ifcs>hithenify2=lcthencd=cs-128:c(x2,y2)=cd:gosub700:x2=n:cs=n
E954 610 ifos>nthenifcs>nthenifx2<>oxthenify2=lcthengosub1150
A81E 620 ifcs>nthengosub1280
956E 630 ox=x2:oy=y2:os=cs:ifrbthenifos>nandox>nthengosub920
365B 640 ifshthenifmvthenmv$=str$(mv):print"{home}{down*9}{white}"spc(35)right$(" "+mv$,4)
5DAD 650 waitj2,g:waitj2,1:goto490
30F1 660 rem *****
816D 670 rem *** subroutines ***
295F 680 rem *****
2E39 690 rem *** draw card on screen
BF38 700 x=x2:y=y2
E7C3 710 tb=x*5:pokes+4,129:u=0
AA30 720 of=of(x):print"{home}"left$(ln$,y+y-of-of)spc(tb)
B2BD 730 ifcd>hithenprintpt$(sgn(y));:pokes+4,0:return
CC36 740 a=int(cd/nc):r=cd-a*nc-1:ifr<0thenr=12
```

```

C6C7 750 f$=str$(r+2):ifr>8thenf$=" "+mid$("jqka",r-8,1)
8B72 760 e$=d$(a):c$=c$(a):g$=mid$(f$+" ",2,2):f$=right$(" "+f$,2)
F438 770 o=0:ifyandx<7theno=1:ifc(x,y-1)>hitheno=2
FE7B 780 ify-of<10thenprinto$(o)v$e$f$o$w$q$g$e$o$w$u$(u);:pokes+4,0:return
BDDC 790 ify-of<11thenprinto$(o)v$e$f$o$w$q$g$e$o$w$;:pokes+4,0:return
E30A 800 ify-of<12thenprinto$(o)v$e$f$o$w$q$;:pokes+4,0:return
6F45 810 printo$(o);:pokes+4,0:return
F395 820 rem *** handle side panel click
F354 830 ifx2=nory2=nthenreturn
06F9 840 ify2>4andy2<8andy<121thenife<hiorz<hithengosub1010:goto880
B9EA 850 ify2>7andy2<11andy<168thengoto1110
5D19 860 ify2>nandy2<4thenifos>nthengosub920
5BEA 870 ify2>10andy2<13thenify>187andx>283andx<303thenpokeese,1:waitj2,16:goto100
62AC 880 pokeese,1:cs=n:x2=n
D0DC 890 return
9061 900 rem *** foundation pile update
D1CD 910 ifrb=0thenreturn
ED8E 920 a=int(os/nc):r=os-a*nc-1:iffp(a)<>roroy<lc(ox)thenreturn
130E 930 ifr<0thenr=12
C1FF 940 pokeese,1:fp(a)=fp(a)+1:f$=str$(r+2):ifr>8thenf$=" "+mid$("jqka",r-8,1)
9E8F 950 g$=mid$(f$+" ",2,2):print"{home}"left$(ln$,a*2+1)spc(35){white}.c$(a)g$
7909 960 ifox=7thengosub1230:os=n:oy=n
236F 970 ifox<7thenlc(ox)=lc(ox)-1:gosub1420:os=n
246F 980 mv=mv+1:ifz=hiande=zthenwn=0:fori=0to6:wn=wn+lc(i):next:ifwn=-7thengoto1520
4C7C 990 return
8C80 1000 rem *** stock pile
0C15 1010 ifrbthenpc=pc+1and3:pokev+36,p(pc):fori=0to6:pokev+40+i,p(pc):next:return
67E7 1020 pokeese,1:forj=1todw
906C 1030 print"{home}"left$(ln$,10)spc(35):os=n:ox=n:u=hi-e-1:ifu>2thenu=2
4EA5 1040 ifz=hithenz=e:printo$(0)pt$u$(u)b$b$b$b$b$b$b$b$;mv=mv+1:return
15C8 1050 pokes+4,129:z=z+1:cd=cd(z):ifz=hithenprintnc$;
2C32 1060 ifz=hi-2orz=hi-1thenu=hi-z-1:printo$(0)pt$u$(u);
AB4C 1070 u=z-e-1:ifu>2thenu=2
55AB 1080 print"{home}"left$(ln$,g)spc(35):y=0:gosub740:u=0:pokes+4,0:ifz<hithennext
E01B 1090 return
35FC 1100 rem *** waste pile
DE66 1110 ifz=eorox=7andoy>8andoy<12orx2=oxandy2=oythencs=os:return
A20B 1120 pokeese,1:pokev+g,peek(v+g)or130:pokev+2,39:pokev+3,178:pokev+14,39
5064 1130 pokev+15,184:pokeese,131:cs=cd(z):x2=7:y2=12:c(7,12)=cs:return
FC4B 1140 rem *** check if selected card goes under clicked card
45CA 1150 ifcs<26andos<26oros>25andcs>25thenreturn
76FE 1160 cv=int(cs/nc):cv=cs-cv*nc:ov=int(os/nc):ov=os-ov*nc
B6EA 1170 ifcv<=ovorcv-ov>1thenreturn
8933 1180 fori=0tolc(ox)-oy:c(x2,lc(x2)+i+1)=c(ox,oy+i):next:lc(x2)=lc(x2)+i-1
8DF1 1190 ifox<7thenlc(ox)=lc(ox)-i:gosub1420
A7D9 1200 ifox=7thengosub1230
DF8F 1210 gosub1480:c(x2,y2+1)=os:lc(x2)=lc(x2)+1:cs=n:x2=n:return
D8B4 1220 rem *** take a card off the discard pile ***
20D3 1230 x=ox:y=8:tb=35:e=e+1:u=z-e-1:ifu>2thenu=2
89B3 1240 ifz=ethencd=cd(z):print"{home}"left$(ln$,g)spc(35)s$b$b$b$b$b$b$b$b$;
FEF5 1250 ifz>ethencd=cd(z-1):gosub720:fori=ztoe+1step-1:cd(i)=cd(i-1):next
752B 1260 return
9545 1270 rem *** show selector frame
E6EF 1280 xp=x2*f+15:yp=(y2-of)*g+50:pokev+2,xp:pokev+3,yp
C997 1290 lc=lc(x2):ifof=0andlc>12thenlc=12
9E2C 1300 dv=lc-y2:sp=0:pokev+14,xp:pokev+15,(lc-of)*g+56
9B60 1310 ifdv>2thenpokev+4,xp:pokev+5,yp+36:sp=4
C71D 1320 ifdv>4thenpokev+6,xp:pokev+7,yp+68:sp=12
0F73 1330 ifdv>6thenpokev+8,xp:pokev+9,yp+100:sp=28
3468 1340 ifdv>8thenpokev+10,xp:pokev+11,yp+132:sp=60
B84E 1350 ifdv>10thenpokev+12,xp:pokev+13,yp+164:sp=124
4DDE 1360 pokeese,131orsp:pokev+g,peek(v+g)and1:return
7233 1370 rem *** redraw whole column (offset change)
9871 1380 x=x2:y=y2:of(x)=9-of(x):tb=x*5:print"{home}"spc(tb)s$;
88A6 1390 fory=0to5:printb$b$b$b$b$b$b$b$;:next:for y=of(x) tolc(x):cd=c(x,y):gosub710:next
75D0 1400 oy=n:return
89D5 1410 rem *** remove selected cards from source column
0AB5 1420 x=ox:y=oy:u=0:tb=x*5:ify=0thenprint"{home}"spc(tb)s$b$b$b$b$b$b$b$b$;:goto1450
DFFD 1430 y=y-1:ify<of(x)thengosub1380:return
7680 1440 cd=c(x,y):gosub720
2120 1450 a=24-(y-of(x))*2-6:ifa<0thenreturn
28B2 1460 fori=0toa:printb$;:next:return
11D8 1470 rem *** add selected cards to the target column
C366 1480 x=x2:for y=y2+1 tolc(x2)+1:cd=c(x,y):gosub710:next:mv=mv+1:return
3AF6 1490 rem *** change card deck color
4237 1500 pc=pc+1and3:pokev+36,p(pc):fori=0to6:pokev+40+i,p(pc):next:return
C3E6 1510 rem *** show statistics
B595 1520 t$=ti$:print"{home}{reverse off}"spc(13){black}congratulations"
3D5A 1530 print"{light green}game time {white}"left$(t$,2)"h "mid$(t$,3,2)"m "right$(t$,2)"s"

```

```

D176 1540 print"{light green}seed number{white}"sd:print"{light green}card moves {white}"mv:mv=0
F51E 1550 print"{light green}draw mode {white}"dw:waitj2,16:pokefe,0:c=12:y=0:x=0:of=0:u=0
AF3F 1560 printtab(34)"{up}"s$:fori=0to12:forj=0to3:cd=c+j*nc:tb=j*5+14:pokes+4,129
237E 1570 print"{home}"left$(ln$,rnd(0)*10+10)spc(rnd(0)*31)
39A7 1580 b=peek(fe):hi=51:ifbthenpokefe,0:hi=n:g$=""
149D 1590 gosub730:print"{home}"left$(ln$,j*2+1)spc(36)c$g$:next:c=c-1:next
35DD 1600 print"{home}{down}"spc(36)"{light gray}{reverse on}";:fori=0to3
3157 1610 print"@ {down*2}{left*2}";:next:return
136F 1620 rem *****
D8EA 1630 rem *** initialization ***
9340 1640 rem *****
E43B 1650 n=-1:f=40:g=16:hi=51:nc=13:msb(1)=256:v=53248:s=54272:j2=56320:se=v+21
3D1A 1660 fb=251:fc=252:fd=253:fe=254:p(0)=14:p(1)=10:p(2)=12:p(3)=4
2C84 1670 ln$="{down*24}":r$="{left*5}{down}":o$="{reverse off}":rd$="{reverse on}{red}"
6C0E 1680 bl$="{reverse on}{black}":c$(0)=rd$:c$(1)=rd$:c$(2)=bl$:c$(3)=bl$
7C06 1690 q$=r$+"{white}."{reverse on}{space*3}{reverse off}"/":d$(0)=rd$+"!"
B412 1700 d$(1)=rd$+"#":d$(2)=bl$+"%":d$(3)=bl$+"&":s$="{space*5}":v$=r$+"{white}."
F820 1710 w$="{white}"/":b$=r$+s$:v$(0)="off":v$(7)="on ":o$(0)="{white}()"*)"
23B6 1720 o$(1)="{white}:{cm f*3}{white}<":sh$(0)="hide":sh$(1)="show"
2B69 1730 o$(2)="{white}:{reverse on}{white}{cm @*3}{reverse off}{white}<"
4C47 1740 u$(0)=r$+"{white}+,,-":u$(1)=r$+"{white}[£££]":u$(2)=r$+"{white}=>>>?"
1370 1750 n$=r$+"."{reverse on}{white}{cm m*3}{reverse off}"+w$:pt$=n$+n$+n$+n$
2C15 1760 pt$(0)=o$(0)+pt$+u$(0):pt$(1)=o$(2)+pt$+u$(0):cd$(1)="{white}1 {light green}card "
2641 1770 cd$(3)="{white}3 {light green}cards":d1$=r$+"{white}."{reverse on}{light gray}@ "
0952 1780 d2$="{white}{reverse off}"/":d3$=r$+"^";:arrow left":d$=d2$+d3$+d1$
BA08 1790 nc$="{light green}()"*)"+r$+"."{space*3}"/"+b$+b$+r$+"."{space*3}"/"+r$+"+,,-"
4048 1800 dimi,j,x,y,x2,y2,cs,ox,oy,os,lc,cd(hi),c(7,18)
6362 1810 pokev+34,15:pokev+35,1:pokes,31:pokes+1,31:fori=0to3:fp(i)=n:next
DA43 1820 pokev+24,peek(v+24)and240or14:pokev+17,peek(v+17)or64:return
2048 1830 rem *** sprite selector frame setup
EFFB 1840 fori=3to57step3:poke896+i,6:poke898+i,3:poke960+i,6:poke962+i,3:next
672C 1850 poke896,3:poke897,255:poke898,254:poke899,7:poke900,255:poke901,255
C3AF 1860 poke1017,7:poke1018,255:poke1019,255:poke1020,3:poke1021,255:poke1022,254
C0D6 1870 pokev+29,254:pokev+23,254:pokev+27,124
27AA 1880 fori=0to6:pokev+40+i,14:next:fori=0to5:poke2041+i,14:next:poke2047,15
7C16 1890 fori=0to6:pokev+2+i+i,255:pokev+3+i+i,52+i*30:next:return
EC89 1900 rem *** copy char rom to ram at $3800 (14336)
94F9 1910 poke56334,peek(56334)and254:poke1,peek(1)and251:poke781,2:poke782,255
13B7 1920 poke90,0:poke91,209:poke88,0:poke89,57:sys41960:poke1,peek(1)or4
E487 1930 poke56334,peek(56334)or1:fori=0to7:reada:poke14336+i,a:next:fori=0to167
036E 1940 reada:poke14552+i,a:next:fori=0to47:reada:poke14800+i,a:next:return
1480 1950 rem *** custom character data
576A 1960 data 0,8,12,126,126,12,8,0,7,7,7,3,4,3,0,0,255,255,255,255,0,255,0,0,224
7334 1970 data 224,224,192,32,192,0,0,7,7,7,7,7,7,224,224,224,0,224,224,224,224
71C4 1980 data 0,0,0,0,0,0,54,127,127,127,62,28,8,0,102,102,0,0,0,0,0,8,28,62
8F59 1990 data 127,62,28,8,0,153,66,36,0,255,255,255,255,8,28,42,127,42,8,28,0,8,28
3A94 2000 data 62,127,127,54,8,0,153,66,36,153,153,36,66,153,0,0,0,0,3,7,7,7,0,0,0,0
C27F 2010 data 255,255,255,255,0,0,0,0,192,224,224,224,7,7,7,3,0,0,0,0,255,255,255
8FFE 2020 data 255,0,0,0,0,224,224,224,192,0,0,0,0,7,7,7,7,7,7,224,224,224,224
CB56 2030 data 224,224,224,224,7,7,7,4,3,7,7,7,255,255,255,0,255,255,255,255,224,224
D2C0 2040 data 224,32,192,224,224,224,7,7,7,3,4,3,4,3,255,255,255,255,0,255,0,255
E845 2050 data 224,224,224,192,32,192,32,192
7AB3 2060 rem *****
9B11 2070 rem *** 1351 mouse setup ***
604A 2080 rem *****
965C 2090 fori=0to204:reada:poke49152+i,a:next
9822 2100 fori=0to62:reada:poke832+i,a:next:v=53248:pokev+36,14:pokev+39,0
813B 2110 pokev,90:pokev+1,190:pokev+16,0:poke2040,13:pokev+21,1:sys49152:return
A13A 2120 rem *** 1351 mouse driver at $c000 ***
7BCC 2130 rem 251=xpos / 252=ypos / 253=xpos high byte / 254=mouse click values
EA74 2140 rem process mouse click(s) then poke254,0 to wait for next click
0147 2150 data 120,169,13,141,20,3,169,192,141,21,3,88,96,165,254,208,21,173,0,220
0C4C 2160 data 73,255,41,17,240,12,133,254,173,0,208,133,251,173,1,208,133,252,169
7D04 2170 data 128,141,0,220,173,25,212,160,0,32,171,192,140,47,192,164,253,208,11
1FCA 2180 data 172,0,208,192,19,208,4,224,0,208,42,24,109,0,208,141,0,208,168,138
C5FC 2190 data 105,0,41,1,77,16,208,141,16,208,41,1,133,253,208,8,192,19,176,13,169
1187 2200 data 19,208,6,192,81,144,5,169,81,141,0,208,173,26,212,160,0,32,171,192
8F5F 2210 data 140,115,192,172,1,208,192,50,208,4,224,0,240,35,192,247,208,4,224,0
902E 2220 data 208,27,56,73,255,109,1,208,201,50,176,8,201,30,144,8,169,50,208,6,201
48FB 2230 data 247,144,2,169,247,141,1,208,76,49,234,140,181,192,141,203,192,162,0
845A 2240 data 56,233,0,41,127,201,64,176,5,74,208,11,138,96,9,192,201,255,240,248
1E72 2250 data 56,106,202,160,0,96
2E2E 2260 rem *** mouse pointer sprite data ***
D85F 2270 data 6,0,0,9,0,0,9,0,0,9,0,0,9,0,0,9,192,0,9,56,0,9,38,0,9,37,0
F7A5 2280 data 233,36,128,152,4,128,136,0,128,72,0,128,40,0,128,32,0,128
AEE4 2290 data 16,0,128,16,1,0,8,1,0,8,1,0,4,2,0,3,252,0

```